<?xml version="1.0" encoding="utf-8" standalone="yes"?>
<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform">

  <xsl:include href="/content/xsl/main.xsl" />

  <xsl:template match="content">
    <div class='wrapper newsreel ui-corner-all'>
      <xsl:apply-templates select="user" />
    </div>
    <input type='hidden' id='content-scripts' value='profile.js' />
  </xsl:template>

  <xsl:template match="content/user">
    <p class='slogan ui-corner-top profile'>
      <span />
      <label>
        <xsl:choose>
          <xsl:when test="@id=//auth/@id">
            Мой профиль
          </xsl:when>
          <xsl:otherwise>
            Профиль пользователя
          </xsl:otherwise>
        </xsl:choose>
      </label>
    </p>
    <div id='user-profile'>
      <table class='user-profile'>
        <tr>
          <td rowspan='{count(contacts/*) + 1}' id='user-photo'>
            <div class='ui-state-default avatar'>
              <xsl:element name="img">
                <xsl:choose>
                  <xsl:when test="@id=//auth/@id">
                    <xsl:attribute name='class'>change-photo</xsl:attribute>
                  </xsl:when>
                  <xsl:otherwise>
                    <xsl:attribute name='style'>cursor:default;</xsl:attribute>
                  </xsl:otherwise>
                </xsl:choose>
                <xsl:attribute name='src'><xsl:value-of select="@avatar" /></xsl:attribute>
                <xsl:attribute name='alt'>Фото</xsl:attribute>
              </xsl:element>
            </div>
            <xsl:if test="@id=//auth/@id">
              <a href='{//urls/base}/user/avatars'>Выбрать аватар</a>
            </xsl:if>
          </td>
          <td class='name'>Имя:</td>
          <td><xsl:value-of select="@name" /></td>
        </tr>
        <xsl:apply-templates select="contacts" />
      </table>
      <xsl:apply-templates select="posts" />
      <div class='wrapper'>
        <div class='user-profile'>
          <xsl:apply-templates select="about" />
          <xsl:apply-templates select="fields" />
          <xsl:apply-templates select="ps" />
        </div>
      </div>
    </div>
  </xsl:template>

  <xsl:template match="contacts">
    <xsl:apply-templates />
  </xsl:template>

  <xsl:template match="email">
    <xsl:call-template name="contact">
      <xsl:with-param name="name">e-mail</xsl:with-param>
      <xsl:with-param name="default">не заполнено</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="phone">
    <xsl:call-template name="contact">
      <xsl:with-param name="name">Контактный телефон</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="icq">
    <xsl:call-template name="contact">
      <xsl:with-param name="name">ICQ UIN</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="jabber">
    <xsl:call-template name="contact">
      <xsl:with-param name="name">Jabber-аккаунт</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="skype">
    <xsl:call-template name="contact">
      <xsl:with-param name="name">Skype-аккаунт</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="twitter">
    <xsl:call-template name="contact">
      <xsl:with-param name="name">Twitter-аккаунт</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="facebook">
    <xsl:call-template name="contact">
      <xsl:with-param name="name">Facebook-аккаунт</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="vkontakte">
    <xsl:call-template name="contact">
      <xsl:with-param name="name">Аккаунт ВКонтакте</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="myworld">
    <xsl:call-template name="contact">
      <xsl:with-param name="name">Аккаунт в Моем Мире</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="livejournal">
    <xsl:call-template name="contact">
      <xsl:with-param name="name">Аккаунт в Живом Журнале</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="contact">
    <xsl:param name="name" />
    <xsl:param name="default">отсутствует</xsl:param>
    <xsl:if test="(.!='') or (//user/@id=//auth/@id)">
      <tr>
        <td class='name'>
          <xsl:if test="//user/@id=//auth/@id">
            <input type='checkbox' title='Видно всем' />
          </xsl:if>
          <xsl:value-of select="$name" />:
        </td>
        <td>
          <div class='contact' id='contact-{name()}'>
            <span data-default='{$default}'>
              <xsl:choose>
                <xsl:when test=".=''">
                  <xsl:attribute name="class">contact-field disabled</xsl:attribute>
                  (<xsl:value-of select="$default" />)
                </xsl:when>
                <xsl:otherwise>
                  <xsl:attribute name="class">contact-field</xsl:attribute>
                  <xsl:choose>
                    <xsl:when test="@href=''">
                      <xsl:value-of select="." />
                    </xsl:when>
                    <xsl:otherwise>
                      <a href='{@href}'><xsl:value-of select="." /></a>
                    </xsl:otherwise>
                  </xsl:choose>
                </xsl:otherwise>
              </xsl:choose>
            </span>
            <xsl:if test="//user/@id=//auth/@id">
              <span class='ui-icon ui-icon-pencil ui-corner-all change-text' title='изменить' data-contact='{name()}' />
            </xsl:if>
          </div>
        </td>
      </tr>
    </xsl:if>
  </xsl:template>

  <xsl:template match="posts">
    <ul id='post-links'>
      <li>
        <a href='{//urls/base}/posts/blog?f=user&#38;v={../@id}' class='link'>
          <xsl:value-of select="concat('блоги (', blog, ')')" />
        </a>
      </li>
      <li>
        <a href='{//urls/base}/posts/news?f=user&#38;v={../@id}' class='link'>
          <xsl:value-of select="concat('новости (', news, ')')" />
        </a>
      </li>
      <li>
        <a href='{//urls/base}/posts/public?f=user&#38;v={../@id}' class='link'>
          <xsl:value-of select="concat('публикации (', public, ')')" />
        </a>
      </li>
    </ul>
  </xsl:template>

  <xsl:template match="about">
    <xsl:if test="(.!='') or (//user/@id=//auth/@id)">
      <div class='name contact'>
        Кратко о себе:
        <xsl:if test="//user/@id=//auth/@id">
          <span class='ui-icon ui-icon-pencil ui-corner-all' title='изменить' />
        </xsl:if>
      </div>
      <div class='field htmlfield'>
        <p>
          <xsl:choose>
            <xsl:when test=".=''">
              <xsl:attribute name="class">text disabled</xsl:attribute>
              (не заполнено)
            </xsl:when>
            <xsl:otherwise>
              <xsl:attribute name="class">text</xsl:attribute>
              <xsl:value-of select="." disable-output-escaping="yes" />
            </xsl:otherwise>
          </xsl:choose>
        </p>
      </div>
    </xsl:if>
  </xsl:template>

  <xsl:template match="fields">
    <table>
      <xsl:apply-templates />
    </table>
  </xsl:template>

  <xsl:template match="sex">
    <xsl:call-template name="field">
      <xsl:with-param name="name">Пол</xsl:with-param>
      <xsl:with-param name="value">
        <xsl:choose>
          <xsl:when test=".='male'">мужской</xsl:when>
          <xsl:when test=".='female'">женский</xsl:when>
        </xsl:choose>
      </xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="frm_birthday">
    <xsl:call-template name="field">
      <xsl:with-param name="name">День рождения</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="hometown">
    <xsl:call-template name="field">
      <xsl:with-param name="name">Родной город</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="now_live">
    <xsl:call-template name="field">
      <xsl:with-param name="name">Сейчас проживаю</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="education">
    <xsl:call-template name="field">
      <xsl:with-param name="name">Образование</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="sstatus">
    <xsl:call-template name="field">
      <xsl:with-param name="name">Социальный статус</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="mstatus">
    <xsl:call-template name="field">
      <xsl:with-param name="name">Семейное положение</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="politics">
    <xsl:call-template name="field">
      <xsl:with-param name="name">Политические убеждения</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template match="religion">
    <xsl:call-template name="field">
      <xsl:with-param name="name">Религиозные взгляды</xsl:with-param>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="field">
    <xsl:param name="name" />
    <xsl:param name="value" />
    <xsl:if test="(.!='') or (//user/@id=//auth/@id)">
      <tr>
        <td class='name'>
          <xsl:value-of select="$name" />:
        </td>
        <td>
          <xsl:choose>
            <xsl:when test="(.='') and ($value='')">
              <xsl:attribute name="class">field disabled</xsl:attribute>
              (не заполнено)
            </xsl:when>
            <xsl:otherwise>
              <xsl:attribute name="class">field</xsl:attribute>
              <xsl:choose>
                <xsl:when test="$value=''">
                  <xsl:value-of select="." />
                </xsl:when>
                <xsl:otherwise>
                  <xsl:value-of select="$value" />
                </xsl:otherwise>
              </xsl:choose>
            </xsl:otherwise>
          </xsl:choose>
          <xsl:if test="//user/@id=//auth/@id">
            <span class='ui-icon ui-icon-pencil ui-corner-all' title='изменить' />
          </xsl:if>
        </td>
      </tr>
    </xsl:if>
  </xsl:template>

  <xsl:template match="ps">
    <xsl:if test="(.!='') or (//user/@id=//auth/@id)">
      <div class='name field'>
        P.S.:
        <xsl:if test="//user/@id=//auth/@id">
          <span class='ui-icon ui-icon-pencil ui-corner-all' title='изменить' />
        </xsl:if>
      </div>
      <div class='field htmlfield'>
        <p>
          <xsl:choose>
            <xsl:when test=".=''">
              <xsl:attribute name="class">text disabled</xsl:attribute>
              (отсутствует)
            </xsl:when>
            <xsl:otherwise>
              <xsl:attribute name="class">text</xsl:attribute>
              <xsl:value-of select="." disable-output-escaping="yes" />
            </xsl:otherwise>
          </xsl:choose>
        </p>
      </div>
    </xsl:if>
  </xsl:template>

</xsl:stylesheet>
